

EXACT PROBABILITIES FOR THREE AND FOUR DICE IN THE BALANCED UNIFORM MODEL OF 3-SIDED DICE

KENT E. MORRISON

ABSTRACT. We determine the exact probabilities of the different isomorphism classes of tournaments that result from random sets of three and four independent dice drawn from the balanced uniform model of 3-sided dice.

1. BACKGROUND AND RESULTS

By now it is well-recognized that various models of generalized dice exhibit surprisingly high rates of intransitivity for randomly chosen dice. Given two dice A and B , where A has face values a_1, \dots, a_n and B has face values b_1, \dots, b_m , we say that A **dominates** B , denoted by $A \succ B$, if it is more likely for A to show a higher value than B , that is

$$\sum_{i,j} \text{sgn}(a_i - b_j) > 0.$$

If the sum is 0, then neither die dominates the other and we say that the result is a tie. In this article we will always be working with dice with the same number of faces; i. e., $m = n$.

In 2016 Conrey, Gabbard, Grant, Liu, and Morrison [1] considered n -sided dice as integer multisets of size n with elements in $\{1, 2, \dots, n\}$ and satisfying $\sum a_i = n(n+1)/2$. We made two conjectures about random sets of three n -sided dice as $n \rightarrow \infty$.

- The probability of any ties goes to zero.
- The probability of an intransitive triple goes to $1/4$.

For three dice A, B, C and in the absence of ties there are 2^3 possible configurations for the relations between the three pairs (A, B) , (B, C) , and (A, C) . Two of the eight configurations represent the intransitive cycles $A \succ B \succ C \succ A$ and $A \succ C \succ B \succ A$. The other six configurations are transitive chains. Evidence for the conjecture was provided by Monte Carlo simulations showing that the probabilities of all eight configurations appear to approach $1/8$, and so the probability of an intransitive triple approaches $1/4$ and the probability of a transitive triple approaches $3/4$. We also made much more speculative

Date: March 8, 2023.

generalized conjectures concerning k random dice for any fixed positive integer k . First, the probability that there is a tie between any of the k dice goes to 0. With no ties there are $2^{\binom{k}{2}}$ outcomes for all pairwise comparison, and each of these outcomes is represented by a complete directed graph on k vertices, i.e., a *tournament*. The second generalized conjecture is that in the limit all the tournaments are equally probable.

A few years later these conjectures became the topic of a Polymath project. An alternative model was introduced, the *balanced sequence model*, in which an n -sided die is a sequence $A = (a_1, \dots, a_n)$ of elements in $\{1, 2, \dots, n\}$ with total $n(n+1)/2$. The balanced sequence model is easier to work than the multiset model. For example, a random die is now a sequence of n iid random variables uniform in $\{1, \dots, n\}$ and conditioned on the sum being $n(n+1)/2$. The main results in [3] are that the conjectures for three dice hold for the balanced sequence model. Thus, the questions for three dice in the multiset model are still open, but it would be a surprise if they turn out to be false.

At the same time the Polymath project spurred some other work, and there is strong computational evidence that the generalized conjecture about tournaments fails with four dice. If the $2^{\binom{4}{2}}$ tournaments were all equally probable, then the probability of a transitive chain and the probability of an intransitive cycle would both have limit $3/8 = 0.375$, but several independent simulations done for both the multiset and the sequence models produce data with the experimental probabilities somewhat greater than 0.38.

Further evidence appears in the results of Cornacchia and Hażła [2] for four dice in the balanced uniform model. In this model a random n -sided die is a point $(a_1, \dots, a_n) \in [0, 1]^n$ chosen uniformly and conditioned on $\sum_i a_i = n/2$. They prove that there exists $\varepsilon > 0$ such that for n sufficiently large, the probability is greater than $3/8 + \varepsilon$ that four random dice have a transitive tournament. In the same paper they prove that, in fact, the conjectures do hold for three dice in the balanced uniform model. That is, as $n \rightarrow \infty$, the probability that three n -sided dice form an intransitive triple approaches $1/4$, and the probability that they form a transitive chain approaches $3/4$.

In this note we consider random sets of three dice and four dice at the other extreme with $n = 3$, which is the least number of sides of any interest. With three dice there are eight tournaments in two isomorphism classes. One isomorphism class consists of the six transitive chains and the other class contains the two intransitive cycles. We find the transitive probability is exactly $973/1230 = 0.76015625$ and the intransitive probability is exactly $307/1280 = 0.23984375$. With four dice the 64 tournaments fall into four different isomorphism classes: the 24 completely transitive chains, the 24 intransitive cycles, the 8 tournaments with an overall winner and a 3-cycle, and the 8 tournaments with an overall

loser and a 3-cycle. We find the probabilities are:

$$\text{transitive} \quad \frac{110413771}{25804800} \approx 0.42788$$

$$\text{intransitive} \quad \frac{99930571}{25804800} \approx 0.38726$$

$$\text{winner + 3-cycle} \quad \frac{23851829}{25804800} \approx 0.09243$$

$$\text{loser + 3-cycle} \quad \frac{23851829}{25804800} \approx 0.09243$$

2. THREE DICE

The dominance relation between A and B is invariant under permutations of the coordinates, and so we can assume that $a_1 \leq a_2 \leq a_3$. We also have $a_1 + a_2 + a_3 = 3/2$. The sample space for a single die is the polygonal region in the $a_1 a_2$ -plane defined by the inequalities

$$0 \leq a_1, \quad a_1 \leq a_2, \quad 1 \leq 2a_1 + 2a_2, \quad 2a_1 + 4a_2 \leq 3.$$

The third and fourth inequalities come from $a_3 \leq 1$ and $a_2 \leq a_3$ by replacing a_3 with $3/2 - a_1 - a_2$. Call this region Q . Its boundary is a quadrilateral within the unit square and it has area $1/8$. The probability measure on Q is the normalized area. See Figure 1.

The sample space for three dice (A, B, C) is the six-dimensional polytope $Q^3 \subset [0, 1]^6$. It is defined by 12 linear inequalities, four for each of the three dice. The volume of Q^3 is $1/8^3 = 1/512$. We use the coordinates $(a_1, a_2, b_1, b_2, c_1, c_2)$ for points in Q^3 .

Our goal is to compute the volume of the subset of Q^3 defined by $A \succ B \succ C \succ A$. Let's first consider the inequalities that define $A \succ B$. It is not necessary to check all 9 comparisons between each a_i and b_j . It is sufficient to simply compare a_i with b_i .


Lemma 1. *There are three different configurations for which $A \succ B$. They are*

$$(\succ_1) : a_1 < b_1, \quad a_2 > b_2, \quad a_3 > b_3$$

$$(\succ_2) : a_1 > b_1, \quad a_2 < b_2, \quad a_3 > b_3$$

$$(\succ_3) : a_1 > b_1, \quad a_2 > b_2, \quad a_3 < b_3$$

Proof. If $a_i > b_i$ for all $i = 1, 2, 3$, then $\sum a_i > \sum b_i$, which is impossible. However, if any two of the three hold, then A dominates B . The configurations are labeled \succ_i according to which inequality $a_i > b_i$ fails to hold. \square

FIGURE 1. The sample space Q for a single random die

Lemma 2. *If $A \succ_i B$ and $B \succ_i C$, then $A \succ_i C$. (That is, the relations \succ_i are transitive.)*

Proof. Straightforward. Just write down the inequalities. □

Lemma 3. *If A, B, C form an intransitive triple with $A \succ B \succ C \succ A$, then $A \succ_i B \succ_j C \succ_k A$ where $\{i, j, k\} = \{1, 2, 3\}$.*

Proof. Assume that the three relations are not distinct. By relabeling, if necessary, we can assume that $i = j$. Then $A \succ_i B$ and $B \succ_i C$, and therefore $A \succ_i C$, which is a contradiction. □

For a permutation $\sigma = \sigma_1\sigma_2\sigma_3$ define E_σ to be both the event

$$A \succ_{\sigma_1} B \succ_{\sigma_2} C \succ_{\sigma_3} A$$

and the subset of Q^3 representing the event. Each E_σ is a polytope defined by nine additional inequalities, three for each of the three relations. The union $E = \bigcup E_\sigma$ is the event that $A \succ B \succ C \succ A$, and $P(E) = \text{vol}(E)/\text{vol}(Q^3)$. Also, $\text{vol}(E) = \sum_\sigma \text{vol}(E_\sigma)$. The volumes of the E_σ are computed using the SageMath interface to LattE integrale.

Lemma 4.

$$P(E_\sigma) = \begin{cases} \frac{23}{1800} & \text{if } \sigma = 123, 231, 312 \\ \frac{3133}{115200} & \text{if } \sigma = 132, 213, 321 \end{cases}$$

$$P(E) = \frac{307}{2560}$$

Proof. See Section 4 for the code. Note that it suffices to compute the volumes of E_{123} and E_{132} because E_σ and $E_{\sigma'}$ have the same volume if σ' is a cyclic permutation of σ . The reason is that the cyclic permutation $(A, B, C) \mapsto (B, C, A)$ is volume preserving and maps $E_{\sigma_1\sigma_2\sigma_3}$ to $E_{\sigma_2\sigma_3\sigma_1}$. □

Theorem 5. *The probability that A, B, C form an intransitive triple is $307/1280 = 0.23984375$. The probability that they form a transitive chain is $973/1280 = 0.76015625$.*

Proof. By symmetry $P(A \succ B \succ C \succ A) = P(A \prec B \prec C \prec A)$, which is $P(E)$, and so the probability of an intransitive triple is $2P(E)$. The transitive probability is the complementary probability $1 - 2P(E)$. □

3. FOUR DICE

The sample space for four dice (A, B, C, D) is the eight-dimensional polytope $Q^4 \subset [0, 1]^8$. It is defined by 16 linear inequalities, four for each of the four dice. The volume of Q^4 is $1/8^4 = 1/4096$. We use the coordinates $(a_1, a_2, b_1, b_2, c_1, c_2, d_1, d_2)$ for points in Q^4 .

The main computation that we need to do is to find the probability that $A \succ B \succ C \succ D \succ A$. Then using symmetry and the results already computed for three dice we will have enough to determine the probabilities of the four different tournament types. To see how that works, we introduce the following notation for the probabilities of the different isomorphism classes for three and four dice.

transitive 3-chain	$P_{3\text{-line}}$
intransitive 3-cycle	P_{Δ}
transitive 4-chain	$P_{4\text{-line}}$
intransitive 4-cycle	P_{\square}
winner + 3-cycle	$P_{1 \rightarrow \Delta}$
loser + 3-cycle	$P_{1 \leftarrow \Delta}$

Generating a random set of four dice and then deleting one of them at random gives a random set of three dice. If the four dice are a transitive chain, then the remaining three dice will be transitive. If the four dice are a 4-cycle, then removing one of them results in a 3-cycle or a 3-chain, with two possibilities for each. If the four dice are a 3-cycle plus winner/loser, then the result is a 3-cycle when the winner/loser is removed or a 3-chain if one of the other three dice is removed. Therefore,

$$P_{3\text{-line}} = P_{4\text{-line}} + \frac{1}{2}P_{\square} + \frac{3}{4}(P_{1 \rightarrow \Delta} + P_{1 \leftarrow \Delta})$$

$$P_{\Delta} = \frac{1}{2}P_{\square} + \frac{1}{4}(P_{1 \rightarrow \Delta} + P_{1 \leftarrow \Delta})$$

Furthermore, $P_{1 \rightarrow \Delta} = P_{1 \leftarrow \Delta}$, because the transformation

$$A = (a_1, a_2, a_3) \mapsto A^* = (1 - a_3, 1 - a_2, 1 - a_1)$$

reverses the relation between dice, and so the tournament associated to (A^*, B^*, C^*, D^*) is the result of reversing all the edges in the tournament of (A, B, C, D) . Therefore, once we have found P_{\square} , we use it and the already known $P_{3\text{-line}}$ and P_{Δ} to solve for the remaining probabilities.

Lemma 6. $P_{\square} = 6 P(A \succ B \succ C \succ D \succ A)$.

Proof. There are six ways to label the vertices of a 4-cycle. □

A priori, there are 81 different polytopes in Q^4 to consider because each of the four relations in the cycle $A \succ B \succ C \succ D \succ A$ is \succ_1, \succ_2 , or \succ_3 . For $\sigma = \sigma_1 \sigma_2 \sigma_3 \sigma_4$, where $\sigma_i \in \{1, 2, 3\}$, let G_{σ} be the event $A \succ_{\sigma_1} B \succ_{\sigma_2} C \succ_{\sigma_3} D \succ_{\sigma_4} A$ as well as the corresponding subset of Q^4 . The union $G = \bigcup G_{\sigma}$ corresponds to the event $A \succ B \succ C \succ D \succ A$, and $P(G) = \sum P(G_{\sigma})$. Fortunately, we can reduce the number of G_{σ} whose volumes need to be computed.

Lemma 7. *If σ' is a cyclic permutation of σ , then $\text{vol}(G_{\sigma'}) = \text{vol}(G_{\sigma})$.*

Proof. The map $Q^4 \rightarrow Q^4 : (A, B, C, D) \mapsto (B, C, D, A)$ is an isometry and maps G_{σ} to $G_{\sigma'}$ where $\sigma = (\sigma_1, \sigma_2, \sigma_3, \sigma_4)$ and $\sigma' = (\sigma_2, \sigma_3, \sigma_4, \sigma_1)$. □

Lemma 8. *If σ has one or two distinct entries, then $G_{\sigma} = \emptyset$.*

Proof. The σ in question are the following and all cyclic permutations of them:

$$(iiii) : 1111, 2222, 3333$$

$$(iiij) : 1112, 1113, 2221, 2223, 3331, 3332$$

$$(iijj) : 1122, 1133, 2233$$

$$(ijij) : 1212, 1313, 2323$$

For σ in the first group the chain of inequalities obviously leads to a contradiction. Now let $\sigma = 1112$, which means that $a_1 < b_1 < c_1 < d_1$ and $d_2 < a_2$. But we also have $a_2 > b_2 > c_2 > d_2$, which is a contradiction. The remaining five σ in the second group are disposed of similarly. From the third group we use the fact that the \succ_i are transitive. If $\sigma = iijj$ then G_σ is the event $A \succ_i B \succ_i C \succ_j D \succ_j A$, and this implies that $A \succ_i C$ and $C \succ_j A$, which is impossible. Finally for $\sigma = ijij$ we consider $A \succ_i B \succ_j C \succ_i D \succ_j A$. Then for $k \neq i, j$ we have $a_k > b_k > c_k > d_k > a_k$, which is impossible. \square

Lemma 9. *There are 36 different σ for which $P(G_\sigma) > 0$. Each is a cyclic permutation of one of the nine on the list below. If σ' is a cyclic permutation of σ , then $P(G_{\sigma'}) = P(G_\sigma)$.*

σ	$P(G_\sigma)$
1123	229/322560
1132	691507/294912000
1213	40913/15482880
1223	5431/8064000
1232	32299/16515072
1322	38929/18432000
1233	229/322560
1323	40913/15482880
1332	691507/294912000

Proof. See the Appendix for the Sage code to compute the volumes and probabilities. \square

Notice that $P(G_{1123}) = P(G_{1233})$, $P(G_{1132}) = P(G_{1332})$, and $P(G_{1213}) = P(G_{1323})$. Consider the event G_{1123} , which means that $A \succ_1 B \succ_1 C \succ_2 D \succ_3 A$. Apply the star operator to each die and verify that $A^* \succ_1 D^* \succ_2 C^* \succ_3 B^* \succ_3 A^*$. That is, \succ_1 reverses direction and changes to \succ_3 , while \succ_3 reverses and changes to \succ_1 , and \succ_2 only reverses direction. Thus, the isometry

$$Q^4 \rightarrow Q^4 : (A, B, C, D) \mapsto (A^*, D^*, C^*, B^*)$$

maps G_{1123} onto G_{1233} . It also maps G_{1132} onto G_{2133} , which has the same volume as G_{1332} because 2133 and 1332 are cyclically related. And it maps G_{1213} onto G_{1323} .

Lemma 10.

$$P(A \succ B \succ C \succ D \succ A) = P(G) = \frac{99930571}{1548288000}$$

Proof. There are four cyclic permutations of each σ in Lemma 9. So summing them and multiplying by four gives $P(G)$. \square

Theorem 11. *With four 3-sided dice in the balanced uniform model, the probabilities of the four isomorphism classes of tournaments are the following:*

$$P_{4\text{-line}} = \frac{110413771}{258048000} \approx 0.42788 \quad (\text{transitive chain})$$

$$P_{\square} = \frac{99930571}{258048000} \approx 0.38726 \quad (\text{intransitive 4-cycle})$$

$$P_{1 \rightarrow \Delta} = \frac{23851829}{258048000} \approx 0.09243 \quad (\text{winner + 3-cycle})$$

$$P_{1 \leftarrow \Delta} = \frac{23851829}{258048000} \approx 0.09243 \quad (\text{loser + 3-cycle})$$

Proof. With four vertices there are six different 4-cycles. From Lemma 9 we know $P(G)$, the probability of one particular 4-cycle, and so

$$P_{\square} = 6 P(G) = 6 \left(\frac{99930571}{1548288000} \right) = \frac{99930571}{258048000}.$$

We also know $P_{3\text{-line}} = 973/1280$ and $P_{\Delta} = 307/1280$ from Theorem 5, so that we can solve for $P_{4\text{-line}}$, $P_{1 \rightarrow \Delta}$, and $P_{1 \leftarrow \Delta}$ using the equations

$$P_{3\text{-line}} = P_{4\text{-line}} + \frac{1}{2}P_{\square} + \frac{3}{4}(P_{1 \rightarrow \Delta} + P_{1 \leftarrow \Delta})$$

$$P_{\Delta} = \frac{1}{2}P_{\square} + \frac{1}{4}(P_{1 \rightarrow \Delta} + P_{1 \leftarrow \Delta})$$

$$P_{1 \rightarrow \Delta} = P_{1 \leftarrow \Delta}$$

\square

4. SAGE COMPUTATION: THREE DICE

We define the polytopes (or “polyhedra”) in Sage using the half-space representation [4], which is a system of linear inequalities of the form

$$0 \leq \beta + \alpha_1 x_1 + \alpha_2 x_2 + \cdots + \alpha_n x_n.$$

This inequality is represented by $(\beta, \alpha_1, \alpha_2, \dots, \alpha_n)$, a tuple of length $n + 1$. Our variables are $a_1, a_2, b_1, b_2, c_1, c_2$, so that, for example, the inequality $2a_1 + 4a_2 \leq 3$ is represented by the 7-tuple $(3, -2, -4, 0, 0, 0, 0)$. The twelve inequalities defining Q^3 and their Sage representations are the following:

$$\begin{array}{llll} 0 \leq a_1, & a_1 \leq a_2, & 1 \leq 2a_1 + 2a_2, & 2a_1 + 4a_2 \leq 3, \\ 0 \leq b_1, & b_1 \leq b_2, & 1 \leq 2b_1 + 2b_2, & 2b_1 + 4b_2 \leq 3, \\ 0 \leq c_1, & c_1 \leq c_2, & 1 \leq 2c_1 + 2c_2, & 2c_1 + 4c_2 \leq 3. \end{array}$$

$$\begin{array}{llll} (0, 1, 0, 0, 0, 0, 0), & (0, -1, 1, 0, 0, 0, 0), & (-1, 2, 2, 0, 0, 0, 0), & (3, -2, -4, 0, 0, 0, 0), \\ (0, 0, 0, 1, 0, 0, 0), & (0, 0, 0, -1, 1, 0, 0), & (-1, 0, 0, 2, 2, 0, 0), & (3, 0, 0, -2, -4, 0, 0), \\ (0, 0, 0, 0, 0, 1, 0), & (0, 0, 0, 0, 0, -1, 1), & (-1, 0, 0, 0, 0, 2, 2), & (3, 0, 0, 0, 0, -2, -4). \end{array}$$

To define E_{123} corresponding to $A \succ_1 B \succ_2 C \succ_3 A$ there are nine additional inequalities.

$$\begin{array}{lll} a_1 < b_1, & a_2 > b_2, & a_3 > b_3 \Leftrightarrow a_1 + a_2 < b_1 + b_2, \\ b_1 > c_1, & b_2 < c_2, & b_3 > c_3 \Leftrightarrow b_1 + b_2 < c_1 + c_2, \\ c_1 > a_1, & c_2 > a_2, & c_3 < a_3 \Leftrightarrow c_1 + c_2 > a_1 + a_2. \end{array}$$

The Sage representations are

$$\begin{array}{lll} (0, -1, 0, 1, 0, 0, 0), & (0, 0, 1, 0, -1, 0, 0), & (0, -1, -1, 1, 1, 0, 0), \\ (0, 0, 0, 1, 0, -1, 0), & (0, 0, 0, 0, -1, 0, 1), & (0, 0, 0, -1, -1, 1, 1), \\ (0, -1, 0, 0, 0, 1, 0), & (0, 0, -1, 0, 0, 0, 1), & (0, -1, -1, 0, 0, 1, 1). \end{array}$$

The polytope E_{132} corresponding to $A \succ_1 B \succ_3 C \succ_2 A$ has the additional inequalities represented by

$$\begin{array}{lll} (0, -1, 0, 1, 0, 0, 0), & (0, 0, 1, 0, -1, 0, 0), & (0, -1, -1, 1, 1, 0, 0), \\ (0, 0, 0, 1, 0, -1, 0), & (0, 0, 0, 0, 1, 0, -1), & (0, 0, 0, 1, 1, -1, -1), \\ (0, -1, 0, 0, 0, 1, 0), & (0, 0, 1, 0, 0, 0, -1), & (0, 1, 1, 0, 0, -1, -1). \end{array}$$

For the volume computations we call the integrate function in *LattE integrale* using the Sage interface [5].

Sage input

```

from sage.interfaces.latte import integrate

Q3_ieqs = [(0,1,0,0,0,0,0), (-1,2,2,0,0,0,0), (0,-1,1,0,0,0,0),
            (3,-2,-4,0,0,0,0), (0,0,0,1,0,0,0), (-1,0,0,2,2,0,0),
            (0,0,0,-1,1,0,0), (3,0,0,-2,-4,0,0), (0,0,0,0,0,1,0),
            (-1,0,0,0,0,2,2), (0,0,0,0,0,-1,1), (3,0,0,0,0,-2,-4)]
Q3 = Polyhedron(ieqs = Q3_ieqs)
vol_Q3 = integrate(Q3.cdd_Hrepresentation(), cdd=True)

E123_ieqs = [(0,-1,0,1,0,0,0), (0,0,1,0,-1,0,0), (0,-1,-1,1,1,0,0),
              (0,0,0,1,0,-1,0), (0,0,0,0,-1,0,1), (0,0,0,-1,-1,1,1),
              (0,-1,0,0,0,1,0), (0,0,-1,0,0,0,1), (0,-1,-1,0,0,1,1)]
E123 = Polyhedron(ieqs = Q3_ieqs + E123_ieqs)
vol_E123 = integrate(E123.cdd_Hrepresentation(), cdd=True)
prob_E123 = vol_E123/vol_Q3

E213_ieqs = [(0,-1,0,1,0,0,0), (0,0,1,0,-1,0,0), (0,-1,-1,1,1,0,0),
              (0,0,0,1,0,-1,0), (0,0,0,0,1,0,-1), (0,0,0,1,1,-1,-1),
              (0,-1,0,0,0,1,0), (0,0,1,0,0,0,-1), (0,1,1,0,0,-1,-1)]
E213 = Polyhedron(ieqs = Q3_ieqs + E213_ieqs)
vol_E213 = integrate(E213.cdd_Hrepresentation(), cdd=True)
prob_E213 = vol_E213/vol_Q3

prob_E = 3*prob_E123 + 3*prob_E213

print("vol(Q3) =", vol_Q3)
print("P(E123) =", prob_E123, "=", n(prob_E123, digits=9))
print("P(E213) =", prob_E213, "=", n(prob_E213, digits=9))
print("P(A>B>C>A) = P(E) =", prob_E, "=", n(prob_E, digits=9))
print("P(intransitive) =", 2*prob_E, "=", n(2*prob_E, digits=9))

```

Output

```

vol(Q3) = 1/512
P(E123) = 23/1800 = 0.0127777778
P(E213) = 3133/115200 = 0.0271961806
P(A>B>C>A) = P(E) = 307/2560 = 0.119921875
P(intransitive) = 307/1280 = 0.239843750

```

5. SAGE COMPUTATION: FOUR DICE

There are 16 inequalities needed to define Q^4 in Sage. Each is represented by a tuple of length 9. They are straightforward extensions of those used to define Q^3 .

To define G_σ where $\sigma = ijkl$ we use the function $G(i, j, k, l)$ which returns the Sage polyhedron. This requires the inequalities for Q^4 and the inequalities for the four relations.

For $m = 1, 2, 3, 4$ and $i = 1, 2, 3$ the function $g(m, i)$ returns a list of three tuples asserting that the m th dominance relation is \succ_i . For example, $g(3, 2)$ returns the tuples for $C \succ_2 D$ since the third relation is the one between C and D . Then $G(i, j, k, l)$ simply concatenates the inequalities for Q^4 with those coming from $g(1, i)$, $g(2, j)$, $g(3, k)$, and $g(4, l)$, a total of 28 inequalities. The definition of $g(m, i)$ uses vectors $a_1, a_2, a_3, \dots, d_1, d_2, d_3$ so that an inequality such as $a_2 > b_2$ is represented by $a_2 - b_2$. The vector a_3 is actually $-a_1 - a_2$ and likewise for b_3, c_3, d_3 , because an inequality such as $a_3 > b_3$ is equivalent to $-a_1 - a_2 > -b_1 - b_2$.

The function $pr(i, j, k, l)$ returns the probability of G_{ijkl} . First it finds the dimension of $G(i, j, k, l)$. If the dimension is less than eight, then the polytope has zero volume and the probability is zero. If the dimension is eight, then the function uses the integrate function from *LattE integrale* to find the volume and multiplies by 4096 to get the probability. It is necessary to check the dimension first, because the integrate function does not return zero on polytopes of less than the full dimension.

Sage input

```

from sage.interfaces.latte import integrate

Q4_ieqs = [(0,1,0,0,0,0,0,0,0), (-1,2,2,0,0,0,0,0,0),
           (0,-1,1,0,0,0,0,0,0), (3,-2,-4,0,0,0,0,0,0),
           (0,0,0,1,0,0,0,0,0), (-1,0,0,2,2,0,0,0,0),
           (0,0,0,-1,1,0,0,0,0), (3,0,0,-2,-4,0,0,0,0),
           (0,0,0,0,0,1,0,0,0), (-1,0,0,0,0,2,2,0,0),
           (0,0,0,0,0,-1,1,0,0), (3,0,0,0,0,-2,-4,0,0),
           (0,0,0,0,0,0,0,1,0), (-1,0,0,0,0,0,0,2,2),
           (0,0,0,0,0,0,0,-1,1), (3,0,0,0,0,0,0,-2,-4)]
Q4 = Polyhedron(ieqs = Q4_ieqs)
vol_Q4 = integrate(Q4.cdd_Hrepresentation(), cdd=True)

a1=vector((0,1,0,0,0,0,0,0,0));
a2=vector((0,0,1,0,0,0,0,0,0));
a3=vector((0,-1,-1,0,0,0,0,0,0));
b1=vector((0,0,0,1,0,0,0,0,0));
b2=vector((0,0,0,0,1,0,0,0,0));
b3=vector((0,0,0,-1,-1,0,0,0,0));
c1=vector((0,0,0,0,0,1,0,0,0));
c2=vector((0,0,0,0,0,0,1,0,0));
c3=vector((0,0,0,0,0,-1,-1,0,0));
d1=vector((0,0,0,0,0,0,0,1,0));
d2=vector((0,0,0,0,0,0,0,0,1));
d3=vector((0,0,0,0,0,0,0,-1,-1));
def g(m,i):
    s = ((-1,1,1), (1,-1,1), (1,1,-1))
    if m==1:
        return [s[i-1][0]*(a1-b1), s[i-1][1]*(a2-b2),
                s[i-1][2]*(a3-b3)]
    elif m==2:
        return [s[i-1][0]*(b1-c1), s[i-1][1]*(b2-c2),
                s[i-1][2]*(b3-c3)]
    elif m==3:
        return [s[i-1][0]*(c1-d1), s[i-1][1]*(c2-d2),
                s[i-1][2]*(c3-d3)]
    elif m==4:
        return [s[i-1][0]*(d1-a1), s[i-1][1]*(d2-a2),
                s[i-1][2]*(d3-a3)]
end

```

Sage input

```

def G(i,j,k,l):
    return Polyhedron(ieqs = Q4_ieqs+g(1,i)+g(2,j)+g(3,k)+g(4,l))
end

def pr(i,j,k,l):
    event = G(i,j,k,l)
    if event.dim() < 8:
        return 0
    else:
        return 4096*integrate(event.cdd_Hrepresentation(),cdd=True)
end

print(" Event          Probability")
print("G(1,1,2,3)      ",pr(1,1,2,3))
print("G(1,1,3,2)      ",pr(1,1,3,2))
print("G(1,2,1,3)      ",pr(1,2,1,3))
print("G(1,2,2,3)      ",pr(1,2,2,3))
print("G(1,2,3,2)      ",pr(1,2,3,2))
print("G(1,3,2,2)      ",pr(1,3,2,2))
print("G(1,2,3,3)      ",pr(1,2,3,3))
print("G(1,3,2,3)      ",pr(1,3,2,3))
print("G(1,3,2,2)      ",pr(1,3,2,2))

```

Output

Event	Probability
G(1,1,2,3)	229/322560
G(1,1,3,2)	691507/294912000
G(1,2,1,3)	40913/15482880
G(1,2,2,3)	5431/8064000
G(1,2,3,2)	32299/16515072
G(1,3,2,2)	38929/18432000
G(1,2,3,3)	229/322560
G(1,3,2,3)	40913/15482880
G(1,3,2,2)	38929/18432000

REFERENCES

- [1] B. Conrey, J. Gabbard, K. Grant, A. Liu, and K. E. Morrison. Intransitive dice. *Mathematics Magazine*, 89(2):133–143, 2016.
- [2] E. Cornacchia and J. Hązła. Intransitive dice tournament is not quasirandom. arXiv:2011.10067, 19 Nov 2020.
- [3] D. H. J. Polymath. The probability that a random triple of dice is intransitive. arXiv:2211.16156, 29 Nov 2022.
- [4] Polyhedra, Sage 9.7 Reference Manual, https://doc.sagemath.org/html/en/reference/discrete_geometry/sage/geometry/polyhedron/constructor.html
- [5] Interface to LattE integrale programs, Sage 9.7 Reference Manual, <https://doc.sagemath.org/html/en/reference/interfaces/sage/interfaces/latte.html>

AMERICAN INSTITUTE OF MATHEMATICS, 600 EAST BROKAW ROAD, SAN JOSE, CA 95112

Email address: morrison@aimath.org